```
> with(OreModules):
> with(OreMorphisms):
> with(Stafford):
> with(linalg):
```

Let us consider the third Weyl algebra $A = A_3(\mathbb{Q})$, where $\mathbb{Q}$ is the field of rational numbers,

```
> A := DefineOreAlgebra(diff=[d[1],x[1]], diff=[d[2],x[2]],
> diff=[d[3],x[3]], polynom=[x[1],x[2],x[3]]):
```

and the left A-module M finitely presented by the following matrix:

```
> R := evalm([[x[2]*d[1]/2,x[2]*d[2]+1,x[2]*d[3]+d[1]/2],
> [-x[2]*d[2]/2-3/2,0,d[2]/2],[-d[1]-x[2]*d[3]/2,-d[2],-d[3]/2]]);
```

$$R := \begin{bmatrix} \frac{1}{2}\,x_2\,d_1 & x_2\,d_2+1 & x_2\,d_3+\frac{1}{2}\,d_1 \\ -\frac{1}{2}\,x_2\,d_2-\frac{3}{2} & 0 & \frac{1}{2}\,d_2 \\ -d_1-\frac{1}{2}\,x_2\,d_3 & -d_2 & -\frac{1}{2}\,d_3 \end{bmatrix}$$

The left A-module M corresponds to a system defining the *infinitesimal transformations of the Lie pseudogroup formed by the contact transformations.*

Let us compute the rank of M:

```
> OreRank(R,A);
```

$$1$$

Thus, we get $rank_A$(M) = 1. Let us now study hom_A(M, A). We first compute ker_A(R.).

```
> Q := Involution(SyzygyModule(Involution(R,A),A),A);
```

$$Q := \begin{bmatrix} -d_2 \\ d_1+x_2\,d_3 \\ -2-x_2\,d_2 \end{bmatrix}$$

We obtain ker_A(R.) = Q Aˆ3. In particular, let us check that R Q = 0:

```
> Mult(R,Q,A);
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

Hence, we get hom_A(M, A) is isomorphic to ker_A(R.) = Q A. A form of M is then defined by means of a right multiple Q ξ of Q, where ξ is an element of A.

A unimodular element of M represented by some row vector λ^* in $A^{1\times 3}$ satisfies λ^* (Q ξ^*) = 1 for a certain element ξ^* of A. Since Q admits a left inverse

```
> T := LeftInverse(Q,A);
```

$$T := \begin{bmatrix} \frac{1}{2}\,x_2 & 0 & \frac{-1}{2} \end{bmatrix}$$

i.e., T Q = 1, if $\lambda^* =$ T, then λ^* represents a unimodular element m* of M and Q induces a left A-homomorphism ϕ from M to A which satisfies $\phi(\mathrm{m}^*) =$ T Q $= 1$.

Serre's Splitting-off theorem cannot be used since $rank_A(\mathrm{M}) < 2$.

```
> UnimodularElement(R,A);
```

```
Error, (in Stafford/UnimodularElementInSubmodule) expecting that the
rank of the left module presented by the first matrix is at least 2.
```

But using the option "checkrank"=false, we can try to detect a unimodular element of M by means of a different method.

```
> U := UnimodularElement(R,A,"checkrank"=false);
```

$$U := \left[\left[\begin{array}{ccc} \frac{1}{2}x_2 & 0 & \frac{-1}{2} \end{array}\right], \left[\begin{array}{c} -d_2 \\ d_1 + x_2\, d_3 \\ -2 - x_2\, d_2 \end{array}\right]\right]$$

We find again the unimodular element m* of M represented by U[1] and the left A-homomorphism ϕ from M to A induced by U[2] which satisfies $\phi(\mathrm{m}^*) = 1$:

```
> Mult(U[1],U[2],A);
```

$$\left[\begin{array}{c} 1 \end{array}\right]$$

Finally, we can check that ϕ is a well-defined left A-homomorphism from M to A since R U[2] $= 0$:

```
> Mult(R,U[2],A);
```

$$\left[\begin{array}{c} 0 \\ 0 \\ 0 \end{array}\right]$$

Now, let us consider a new left A-module M finitely presented by the following matrix:

```
> R := evalm([[d[1]+x[2],d[2],d[3]+x[1]]]);
```

$$R := \left[\begin{array}{ccc} d_1 + x_2 & d_2 & d_3 + x_1 \end{array}\right]$$

The rank of M is clearly 2. Let us compute a unimodular element based on Serre's Splitting-off theorem.

```
> U := UnimodularElement(R,A);
```

$$U := \left[\left[\begin{array}{ccc} 0 & 1 & 0 \end{array}\right], \left[\begin{array}{c} -(d_3 + x_1 + d_2)\,(d_3 + x_1) \\ 1 \\ 2 + d_1\, d_3 + x_1\, d_1 + x_2\, d_3 + x_1\, x_2 + d_2\, d_1 + x_2\, d_2 \end{array}\right]\right]$$

We obtain a unimodular element m* of M represented by U[1] and a left A-homomorphism ϕ from M to A induced by U[2] which satisfies $\phi(\mathrm{m}^*) = 1$:

```
> Mult(U[1],U[2],A);
```

$$\left[\begin{array}{c} 1 \end{array}\right]$$

Let us check that ϕ is a well-defined left A-homomorphism from M to A, i.e., R U[2] $= 0$:

```
> Mult(R,U[2],A);
```

$$\begin{bmatrix} 0 \end{bmatrix}$$

Since M admits a unimodular element, M can be decomposed as a direct sum of A and another left A-module M' up to isomorphism. A presentation of M' can be obtained using the command *FreeDirectSummand* with the option "presentation".

```
> F := FreeDirectSummand(R,A,"presentation"):
> nops(F);
```

$$4$$

The output of the command *FreeDirectSummand* with the option "presentation" is a list with four entries. The first one F[1] is a representative of a unimodular element of M

```
> F[1];
```

$$\begin{bmatrix} 0 & 1 & 0 \end{bmatrix}$$

i.e., m* represented by F[1] is a unimodular element of M. The second entry of F

```
> F[2];
```

$$\begin{bmatrix} -(d_3 + x_1 + d_2)(d_3 + x_1) \\ 1 \\ 2 + d_1 d_3 + x_1 d_1 + x_2 d_3 + x_1 x_2 + d_2 d_1 + x_2 d_2 \end{bmatrix}$$

induces a left A-homomorphism ϕ from M to A which satisfies $\phi(\mathrm{m}^*) = 1$. The third entry of F, namely,

```
> F[3];
```

$$\begin{bmatrix} 1 & 0 & 0 \\ -1 & d_1 + x_2 & -d_3 - x_1 \end{bmatrix}$$

is a presentation matrix of M' which is isomorphic to $\ker(\phi)$. Finally, the last entry of F, namely,

```
> F[4];
```

$$\begin{bmatrix} d_1 + x_2 & d_2 & d_3 + x_1 \\ 1 & d_2 d_3 + {d_3}^2 + {x_1}^2 + 2 d_3 x_1 + d_2 x_1 & 0 \\ 0 & 2 + d_1 d_3 + x_1 d_1 + x_2 d_3 + x_1 x_2 + d_2 d_1 + x_2 d_2 & -1 \end{bmatrix}$$

induces an injective left A-homomorphism i from M' to M. Using the OreMorphism package, we can check again that i is injective:

```
> TestInj(F[3],R,F[4],A);
```

true

Let us compute $rank_A$(M'):

```
> OreRank(F[3],A);
```

$$1$$

Since $rank_A$(M') = 1, we cannot use Serre's Splitting-off theorem again to decompose the left A-module M'.

Using the option "isomorphism" of the command *FreeDirectSummand*,

```
>  G := FreeDirectSummand(R,A,"isomorphism"):
```

we can get another representation of the above splitting. The output G contains

```
>  nops(G);
```

$$4$$

four entries. The first one

```
>  G[1];
```

$$\begin{bmatrix} 0 & 1 & 0 \end{bmatrix}$$

represents the unimodular element m* of M. The second one, namely,

```
>  G[2];
```

$$\begin{bmatrix} -(d_3+x_1+d_2)\,(d_3+x_1) \\ 1 \\ 2+d_1\,d_3+x_1\,d_1+x_2\,d_3+x_1\,x_2+d_2\,d_1+x_2\,d_2 \end{bmatrix}$$

induces a left A-homomorphism ϕ from M to A such that $\phi(\text{m}^*) = 1$. The third one

```
>  G[3];
```

$$\begin{bmatrix} 0 & 1 & 0 & 0 \\ 0 & -1 & d_1+x_2 & -d_3-x_1 \end{bmatrix}$$

is a presentation matrix of the left A-module M_1 which is isomorphic to the direct sum of A and M', and isomorphic to M. Indeed, we note that G[3] = (0 F[3]). The last entry of G

```
>  G[4];
```

$$\begin{bmatrix} 0 & 1 & 0 \\ d_1+x_2 & d_2 & d_3+x_1 \\ 1 & d_2\,d_3+{d_3}^2+{x_1}^2+2\,d_3\,x_1+d_2\,x_1 & 0 \\ 0 & 2+d_1\,d_3+x_1\,d_1+x_2\,d_3+x_1\,x_2+d_2\,d_1+x_2\,d_2 & -1 \end{bmatrix}$$

induces a left A-isomorphism g from M_1 to M. This last result can be checked again using the OreMorphisms package:

```
>  TestIso(G[3],R,G[4],A);
```

$$\mathit{true}$$

We can simplify the presentation G[3] of M_1. Indeed, M_1 is isomorphic to M_2 which is finitely presented by the second entry of S defined by:

```
>  with(PurityFiltration):
>  S := ReducedPresentation(G[3],A);
```

$$S := \left[\begin{bmatrix} 0 & 1 & 0 & 0 \\ 0 & -1 & d_1+x_2 & -d_3-x_1 \end{bmatrix}, \begin{bmatrix} 0 & d_1+x_2 & -d_3-x_1 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}\right]$$

The left A-isomorphism h from M_2 to M_1 is induced by S[4].

```
>   TestIso(S[2],S[1],S[4],A);
```

$$\mathit{true}$$

Hence, if we define P = S[4] G[4], namely,

```
>   P := Mult(S[4],G[4],A);
```

$$P := \begin{bmatrix} 0 & 1 & 0 \\ 1 & d_2\,d_3 + {d_3}^2 + {x_1}^2 + 2\,d_3\,x_1 + d_2\,x_1 & 0 \\ 0 & 2 + d_1\,d_3 + x_1\,d_1 + x_2\,d_3 + x_1\,x_2 + d_2\,d_1 + x_2\,d_2 & -1 \end{bmatrix}$$

then the composition i of g and h, which is induced by P, is a left A-isomorphism.

```
>   TestIso(S[2],R,P,A);
```

$$\mathit{true}$$

Let us now compute i^{-1}.

```
>   Q := InverseMorphism(S[2],R,P,A);
```

$$Q := \left[\begin{bmatrix} -d_2\,d_3 - {d_3}^2 - {x_1}^2 - 2\,d_3\,x_1 - d_2\,x_1 & 1 & 0 \\ 1 & 0 & 0 \\ 2 + d_1\,d_3 + x_1\,d_1 + x_2\,d_3 + x_1\,x_2 + d_2\,d_1 + x_2\,d_2 & 0 & -1 \end{bmatrix}, \begin{bmatrix} 1 \end{bmatrix} \right]$$

We obtain that i^{-1} is induced by Q[1]. Finally, let us check again that i^{-1} is a well-defined left A-isomorphism between M and M_2:

```
>   TestIso(R,S[2],Q[1],A);
```

$$\mathit{true}$$